
NAG Library Routine Document

C09ABF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

C09ABF returns the details of the chosen two-dimensional discrete wavelet filter. For a chosen mother
wavelet, discrete wavelet transform type (single-level or multi-level DWT) and end extension method,
this routine returns the maximum number of levels of resolution (appropriate to a multi-level
transform), the filter length, the total number of approximation, horizontal, vertical and diagonal
coefficients and the number of coefficients in the second dimension for the single-level case. This
routine must be called before any of the two-dimensional transform routines in this chapter.

2 Specification

SUBROUTINE C09ABF (WAVNAM, WTRANS, MODE, M, N, NWLMAX, NF, NWCT, NWCN,
ICOMM, IFAIL)

&

INTEGER M, N, NWLMAX, NF, NWCT, NWCN, ICOMM(180), IFAIL
CHARACTER(*) WAVNAM
CHARACTER(1) WTRANS, MODE

3 Description

Two-dimensional discrete wavelet transforms (DWT) are characterised by the mother wavelet, the end
extension method and whether multiresolution analysis is to be performed. For the selected combination
of choices for these three characteristics, and for given dimensions (m� n) of data matrix A, C09ABF
returns the dimension details for the transform determined by this combination. The dimension details
are: lmax , the maximum number of levels of resolution that would be computed were a multi-level DWT
applied; nf , the filter length; nct the total number of approximation, horizontal, vertical and diagonal
coefficients (over all levels in the multi-level DWT case); and ncn, the number of coefficients in the
second dimension for a single-level DWT. These values are also stored in the communication array
ICOMM, as are the input choices, so that they may be conveniently communicated to the two-
dimensional transform routines in this chapter.

4 References

None.

5 Arguments

1: WAVNAM – CHARACTER(*) Input

On entry: the name of the mother wavelet. See the C09 Chapter Introduction for details.

WAVNAM ¼ HAAR
Haar wavelet.

WAVNAM ¼ 0DBn0, where n ¼ 2; 3; . . . ; 10
Daubechies wavelet with n vanishing moments (2n coefficients). For example,
WAVNAM ¼ DB4 is the name for the Daubechies wavelet with 4 vanishing moments
(8 coefficients).

WAVNAM ¼ 0BIORx.y0, where x.y can be one of 1.1, 1.3, 1.5, 2.2, 2.4, 2.6, 2.8, 3.1, 3.3, 3.5 or
3.7

Biorthogonal wavelet of order x.y. For example WAVNAM ¼ BIOR3:1 is the name for
the biorthogonal wavelet of order 3:1.

Constraint: WAVNAM ¼ HAAR , DB2 , DB3 , DB4 , DB5 , DB6 , DB7 , DB8 , DB9 ,
DB10 , BIOR1:1 , BIOR1:3 , BIOR1:5 , BIOR2:2 , BIOR2:4 , BIOR2:6 , BIOR2:8 , BIOR3:1 ,
BIOR3:3 , BIOR3:5 or BIOR3:7 .

2: WTRANS – CHARACTER(1) Input

On entry: the type of discrete wavelet transform that is to be applied.

WTRANS ¼ S
Single-level decomposition or reconstruction by discrete wavelet transform.

WTRANS ¼ M
Multiresolution, by a multi-level DWT or its inverse.

Constraint: WTRANS ¼ S or M.

3: MODE – CHARACTER(1) Input

On entry: the end extension method.

MODE ¼ P
Periodic end extension.

MODE ¼ H
Half-point symmetric end extension.

MODE ¼ W
Whole-point symmetric end extension.

MODE ¼ Z
Zero end extension.

Constraint: MODE ¼ P , H , W or Z .

4: M – INTEGER Input

On entry: the number of elements, m, in the first dimension (number of rows of data matrix A) of
the input data.

Constraint: M � 2.

5: N – INTEGER Input

On entry: the number of elements, n, in the second dimension (number of columns of data matrix
A) of the input data.

Constraint: N � 2.

6: NWLMAX – INTEGER Output

On exit: the maximum number of levels of resolution, lmax , that can be computed if a multi-level
d i s c r e t e wav e l e t t r a n s f o rm i s a pp l i e d (WTRANS ¼ M ) . I t i s s u c h t h a t
2lmax � min m;nð Þ < 2lmaxþ1, for lmax an integer.

If WTRANS ¼ S , NWLMAX is not set.

7: NF – INTEGER Output

On exit: the filter length, nf , for the supplied mother wavelet. This is used to determine the
number of coefficients to be generated by the chosen transform.

8: NWCT – INTEGER Output

On exit: the total number of wavelet coefficients, nct, that will be generated. When
WTRANS ¼ S the number of rows required in each of the output coefficient matrices can be
calculated as ncm ¼ nct= 4ncnð Þ. When WTRANS ¼ M the length of the array used to store all of
the coefficient matrices must be at least nct.

9: NWCN – INTEGER Output

On exit: for a single-level transform (WTRANS ¼ S ), the number of coefficients that would be
generated in the second dimension, ncn, for each coefficient type. For a multi-level transform
(WTRANS ¼ M ) this is set to 1.

10: ICOMMð180Þ – INTEGER array Communication Array

On exit: contains details of the wavelet transform and the problem dimension which is to be
communicated to the two-dimensional discrete transform routines in this chapter.

11: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, WAVNAM ¼ valueh i was an illegal value.

IFAIL ¼ 2

On entry, WTRANS ¼ valueh i was an illegal value.

IFAIL ¼ 3

On entry, MODE ¼ valueh i was an illegal value.

IFAIL ¼ 4

On entry, M ¼ valueh i.
Constraint: M � 2.

On entry, N ¼ valueh i.
Constraint: N � 2.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

C09ABF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example computes the two-dimensional multi-level resolution for a 6� 6 matrix by a discrete
wavelet transform using the Haar wavelet with whole-point symmetric end extensions. The number of
levels of transformation actually performed is one less than the maximum possible. This number of
levels, the length of the wavelet filter, the total number of coefficients and the number of coefficients in
each dimension for each level are printed along with the vertical detail coefficients from the first level,
before a reconstruction is performed.

10.1 Program Text

Program c09abfe

! C09ABF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: c09abf, c09ecf, c09edf, c09eyf, nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Integer :: i, ifail, lda, ldb, lenc, m, n, nf, &

nwcm, nwcn, nwct, nwl, nwlmax, &
want_coeffs, want_level

Character (10) :: mode, wavnam, wtrans
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: a(:,:), b(:,:), c(:), d(:,:)
Integer, Allocatable :: dwtlevm(:), dwtlevn(:)
Integer :: icomm(180)

! .. Intrinsic Procedures ..
Intrinsic :: sum

! .. Executable Statements ..
Continue
Write (nout,*) ’C09ABF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)

! Read problem parameters
Read (nin,*) m, n

lda = m
ldb = m
Read (nin,*) wavnam, mode
Allocate (a(lda,n),b(ldb,n))

Write (nout,99999) wavnam, mode, m, n

! Read data array and write it out
Do i = 1, m

Read (nin,*) a(i,1:n)
End Do

Write (nout,*) ’ Input Data A :’
Do i = 1, m

Write (nout,99998) a(i,1:n)
End Do

! Query wavelet filter dimensions
! For Multi-Resolution Analysis, decomposition, wtrans = ’M’

wtrans = ’Multilevel’

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call c09abf(wavnam,wtrans,mode,m,n,nwlmax,nf,nwct,nwcn,icomm,ifail)

lenc = nwct
Allocate (c(lenc),dwtlevm(nwlmax),dwtlevn(nwlmax))

! Calculate one less than the max possible number of levels
nwl = nwlmax - 1

! Perform Discrete Wavelet transform
ifail = 0
Call c09ecf(m,n,a,lda,lenc,c,nwl,dwtlevm,dwtlevn,icomm,ifail)

! c09abf returns nwct based on max levels, so recalculate.
nwct = sum(3*dwtlevm(1:nwl)*dwtlevn(1:nwl))
nwct = nwct + dwtlevm(1)*dwtlevn(1)

! Print the details of the transform.
Write (nout,*)
Write (nout,99997) nf
Write (nout,99996) nwl
Write (nout,99995)
Write (nout,99994) dwtlevm(1:nwl)
Write (nout,99993)
Write (nout,99994) dwtlevn(1:nwl)
Write (nout,99992) nwct
Write (nout,*)
Write (nout,99991)
Write (nout,99998) c(1:nwct)

! Now select a nominated matrix of coefficients at a nominated level.
! Remember that level 0 is input data, 1 first coeffs and so on up to nwl,
! which is the deepest level and contains approx. coefficients.

want_level = nwl - 1
! Print only vertical detail coeffs at selected level.

want_coeffs = 1
nwcm = dwtlevm(nwl-want_level+1)
nwcn = dwtlevn(nwl-want_level+1)
Allocate (d(nwcm,nwcn))

! Extract the selected set of coefficients.
Call c09eyf(want_level,want_coeffs,lenc,c,d,nwcm,icomm,ifail)

! Print out the selected coefficients
Write (nout,*)
Write (nout,99989) want_coeffs, want_level
Do i = 1, nwcm

Write (nout,99998) d(i,1:nwcn)

End Do

! Reconstruct original data
ifail = 0
Call c09edf(nwl,lenc,c,m,n,b,ldb,icomm,ifail)

Write (nout,*)
Write (nout,99990)
Do i = 1, m

Write (nout,99998) b(i,1:n)
End Do

99999 Format (1X,’ Parameters read from file :: ’,/,’ Wavelet : ’,A10, &
’ End mode : ’,A10,’ M = ’,I10,’ N = ’,I10)

99998 Format (8(F8.4,1X),:)
99997 Format (1X,’ Length of wavelet filter : ’,I10)
99996 Format (1X,’ Number of Levels : ’,I10)
99995 Format (1X, &

’ Number of coefficients in first dimension for each level : ’)
99994 Format (16X,8(I8,1X),:)
99993 Format (1X, &

’ Number of coefficients in second dimension for each level : ’)
99992 Format (1X,’ Total number of wavelet coefficients : ’,I10)
99991 Format (1X,’ All Wavelet coefficients C : ’)
99990 Format (1X,’ Reconstruction B : ’)
99989 Format (1X,’ Type ’,I1,’ coefficients at selected wavelet level, ’,I4, &

’: ’)
End Program c09abfe

10.2 Program Data

C09ABF Example Program Data
8, 8 : m, n
Haar Whole : wavnam, mode
6.0000 7.0000 8.0000 0.0000 1.0000 9.0000 7.0000 8.0000
9.0000 1.0000 9.0000 9.0000 2.0000 8.0000 1.0000 9.0000
3.0000 0.0000 4.0000 1.0000 3.0000 1.0000 0.0000 4.0000
2.0000 5.0000 9.0000 4.0000 4.0000 2.0000 5.0000 9.0000
1.0000 8.0000 3.0000 3.0000 5.0000 3.0000 8.0000 3.0000
8.0000 1.0000 6.0000 4.0000 6.0000 1.0000 1.0000 6.0000
8.0000 1.0000 1.0000 1.0000 2.0000 3.0000 1.0000 6.0000
9.0000 2.0000 2.0000 4.0000 6.0000 1.0000 2.0000 9.0000

10.3 Program Results

C09ABF Example Program Results

Parameters read from file ::
Wavelet : Haar End mode : Whole M = 8 N = 8

Input Data A :
6.0000 7.0000 8.0000 0.0000 1.0000 9.0000 7.0000 8.0000
9.0000 1.0000 9.0000 9.0000 2.0000 8.0000 1.0000 9.0000
3.0000 0.0000 4.0000 1.0000 3.0000 1.0000 0.0000 4.0000
2.0000 5.0000 9.0000 4.0000 4.0000 2.0000 5.0000 9.0000
1.0000 8.0000 3.0000 3.0000 5.0000 3.0000 8.0000 3.0000
8.0000 1.0000 6.0000 4.0000 6.0000 1.0000 1.0000 6.0000
8.0000 1.0000 1.0000 1.0000 2.0000 3.0000 1.0000 6.0000
9.0000 2.0000 2.0000 4.0000 6.0000 1.0000 2.0000 9.0000

Length of wavelet filter : 2
Number of Levels : 2
Number of coefficients in first dimension for each level :

2 4
Number of coefficients in second dimension for each level :

2 4
Total number of wavelet coefficients : 64

All Wavelet coefficients C :
19.2500 15.5000 18.2500 15.7500 -2.7500 3.5000 -3.2500 -2.2500
5.2500 1.5000 4.2500 0.7500 1.2500 -2.5000 0.7500 0.7500

3.5000 0.0000 0.0000 7.0000 4.0000 4.0000 1.0000 -1.0000
-7.0000 2.0000 3.5000 2.0000 -4.5000 -4.0000 0.0000 -6.0000
1.5000 -2.0000 0.0000 -1.0000 -5.0000 -4.0000 -2.0000 -2.0000
0.0000 -1.0000 0.5000 -1.0000 2.5000 -5.0000 2.0000 -2.0000

-4.5000 3.0000 -7.0000 -0.0000 4.0000 -1.0000 -1.0000 1.0000
-1.0000 0.0000 -1.5000 -3.0000 3.5000 0.0000 5.0000 1.0000

Type 1 coefficients at selected wavelet level, 1:
3.5000 4.0000 -7.0000 -4.5000
0.0000 4.0000 2.0000 -4.0000
0.0000 1.0000 3.5000 0.0000
7.0000 -1.0000 2.0000 -6.0000

Reconstruction B :
6.0000 7.0000 8.0000 0.0000 1.0000 9.0000 7.0000 8.0000
9.0000 1.0000 9.0000 9.0000 2.0000 8.0000 1.0000 9.0000
3.0000 0.0000 4.0000 1.0000 3.0000 1.0000 0.0000 4.0000
2.0000 5.0000 9.0000 4.0000 4.0000 2.0000 5.0000 9.0000
1.0000 8.0000 3.0000 3.0000 5.0000 3.0000 8.0000 3.0000
8.0000 1.0000 6.0000 4.0000 6.0000 1.0000 1.0000 6.0000
8.0000 1.0000 1.0000 1.0000 2.0000 3.0000 1.0000 6.0000
9.0000 2.0000 2.0000 4.0000 6.0000 1.0000 2.0000 9.0000

